# $Id$
# Maintainer: Antonio Rojas <arojas@archlinux.org>
# Contributor: Andrea Scarpino <andrea@archlinux.org>

pkgname=kdevelop-php
pkgver=5.1.80
pkgrel=1
pkgdesc="PHP language and documentation plugin for KDevelop"
arch=(i686 x86_64)
url="http://www.kdevelop.org/"
license=(GPL)
depends=(kdevelop threadweaver kcmutils knewstuff)
makedepends=(extra-cmake-modules kdoctools kdevelop-pg-qt python)
source=("https://download.kde.org/unstable/kdevelop/$pkgver/src/kdev-php-$pkgver.tar.xz"{,.asc})
sha1sums=('7f745fe3ad418f95e1666151f4782196862b943f'
          'SKIP')
validpgpkeys=('329FD02C5AA48FCC77A4BBF0AC44AC6DB29779E6') # Sven Brauch <mail@svenbrauch.de>

prepare() {
  mkdir -p build
}

build() {
  cd build
  cmake ../kdev-php-$pkgver \
    -DCMAKE_INSTALL_PREFIX=/usr \
    -DCMAKE_BUILD_TYPE=Release \
    -DCMAKE_INSTALL_LIBDIR=lib \
    -DBUILD_TESTING=OFF
  make
}

package() {
  cd build
  make DESTDIR="$pkgdir" install
}
